This notebook reproduces some results in the papers:
[1] C. L. Pekeris: "Ground State of Two-Electron Atoms", Phys. Rev. 112(5) (1958), 1649-1658.
[2] C. L. Pekeris: "1 ^{1}S and 2 ^{3}S States of Helium", Phys. Rev. 115(5) (1959), 1216-1221.

Some formulas from the paper [1]

Some steps make use of our package HolonomicFunctions, that can be downloaded from:
http://www.risc.uni- linz.ac.at/research/combinat/software/

```
<< HolonomicFunctions.m
```

HolonomicFunctions package by Christoph Koutschan, RISC-Linz, Version 1.3 (25.01.2010)
⟶ Type ?HolonomicFunctions for help

The Schrödinger differential equation, formula (1). Note that E here is not Euler's constant.

```
pek1[n_] :=
 ToOrePolynomial[Sum[Der[x[i]]^2 + Der[y[i]]^2, {i, n}] + 2 * (E + Z / Sqrt[Sum[x[i]^2, {i, n}]] +
      Z / Sqrt[Sum[y[i]^2, {i, n}]] - 1 / Sqrt[Sum[(y[i] - x[i])^2, {i, n}]]),
  OreAlgebra @@ Join[Table[Der[x[i]], {i, n}], Table[Der[y[i]], {i, n}]]]
```

Formula (5):

```
pek5 = ToOrePolynomial[
   Der[r1]^2 + (2 / r1) ** Der[r1] +
    Der[r2]^2 + (2 / r2) ** Der[r2] + 2 * Der[r12]^2 + (4 / r12) ** Der[r12]
    + ((r1^2 - r2^2 + r12^2) / (r1 * r12)) ** Der[r1] ** Der[r12]
    + ((r2^2 - r1^2 + r12^2) / (r2 * r12)) ** Der[r2] ** Der[r12]
    + 2 * (E + Z / r1 + Z / r2 - 1 / r12),
   OreAlgebra[Der[r1], Der[r2], Der[r12]]];
```

Formula (14) (ϵ is represented by e):

```
pek14 = ToOrePolynomial[e * ((4 u^2 * v + 4 u * v^2 + 4 u^2 * w + 4 u * v * w + 2 * u * w^2) ** Der[u]^2 +
        (4 u^2 * v + 4 u * v^2 + 4 v^2 * w + 4 u * v * w + 2 v * w^2) ** Der[v]^2 +
        (8 u^2 * w + 8 v^2 * w + 4 u * w^2 + 4 v * w^2) ** Der[w]^2 -
        (4 u * w * (2 u + w)) ** Der[u] ** Der[w] - (4 v * w * (2 v + w)) ** Der[v] ** Der[w] +
        (-4 u^2 + 4 v^2 + 2 w^2 + 4 u * w + 4 v * w + 8 u * v - 4 u * v * w - 4 u * v^2 - 4 u^2 * v) ** Der[u] +
        (4 u^2 - 4 v^2 + 2 w^2 + 4 u * w + 4 v * w + 8 u * v - 4 u * v * w - 4 u * v^2 - 4 u^2 * v) ** Der[v] +
        (8 u^2 + 8 v^2 - 4 w^2 - 2 u * w^2 - 2 v * w^2 - 4 u^2 * w - 4 v^2 * w) ** Der[w] -
        4 * (u + v) * (u + v + w)) + 4 * Z * (u + v) * (u + v + w) - (2 u + w) * (2 v + w),
   OreAlgebra[Der[u], Der[v], Der[w]]];
```

Derive recurrence (22) in [1]

The starting point is the time-independent Schrödinger equation

```
ApplyOreOperator[pek1[3], ψ[x[1], x[2], x[3], y[1], y[2], y[3]]] // TraditionalForm
```

$$\left(-\frac{2}{\sqrt{(y(1)-x(1))^2+(y(2)-x(2))^2+(y(3)-x(3))^2}}+\frac{2Z}{\sqrt{x(1)^2+x(2)^2+x(3)^2}}+\frac{2Z}{\sqrt{y(1)^2+y(2)^2+y(3)^2}}+2\,e\right)$$
$$\psi(x(1),x(2),x(3),y(1),y(2),y(3))+\psi^{(0,0,0,0,0,2)}(x(1),x(2),x(3),y(1),y(2),y(3))+\psi^{(0,0,0,0,2,0)}(x(1),x(2),x(3),y(1),y(2),y(3))+$$
$$\psi^{(0,0,0,2,0,0)}(x(1),x(2),x(3),y(1),y(2),y(3))+\psi^{(0,0,2,0,0,0)}(x(1),x(2),x(3),y(1),y(2),y(3))+$$
$$\psi^{(0,2,0,0,0,0)}(x(1),x(2),x(3),y(1),y(2),y(3))+\psi^{(2,0,0,0,0,0)}(x(1),x(2),x(3),y(1),y(2),y(3))$$

We write a little procedure that performs algebraic substitution in a differential equation.
Unfortunately, the command DFiniteSubstitute from our package cannot be applied, since the differential equation that we start with does not have polynomial coefficients.

```
DifferentialSubstitute[op1_, subs : {__Equal}] :=
  Module[{op = Flatten[{op1}], a1, a2, eqs, psi, xv, uv, vars, aes, gb},
   op = ToOrePolynomial[op];
   xv = First /@ First[OreAlgebra[op]]; (* the original variables *)
   uv = First /@ subs;                  (* the variables after substitution *)
   vars = Join[xv, uv];

   (* Get the algebraic equations for the substitutions. *)
   aes = HolonomicFunctions`Private`AlgebraicEquation[Last[#], xv, First[#]] & /@ subs;
   gb = GroebnerBasis[aes, vars, MonomialOrder → Lexicographic];

   (* Do some rewriting with the equations,
   replacing old variables by new ones. *)eqs = ApplyOreOperator[op, psi @@ (Last /@ subs)];
   eqs = eqs /. (If[Head[#[[2]]] === Power,
        #[[2, 1]]^a1_. → #[[1]]^(a1 / #[[2, 2]]), #[[2]] → #[[1]]] & /@ subs);
   eqs = NormalizeCoefficients /@ Together[ToOrePolynomial[eqs,
       psi @@ uv, OreAlgebra @@ (Der /@ uv)]];

   (* Reduce all coefficients with the aes Groebner basis. *)
   eqs = Function[opoly, OrePolynomial[Transpose[{Together[Last[PolynomialReduce[#, gb, vars,
             MonomialOrder → Lexicographic]] & /@ OrePolynomialListCoefficients[opoly]],
         Last[Transpose[First[opoly]]]}], opoly[[2]], opoly[[3]]]] /@ eqs;
   Return[If[Head[op1] === List, #, First[#]] &[Expand[NormalizeCoefficients /@ eqs]]];
  ];
```

With help of this procedure, the change of variables {x1, ..., y3} -> {r1, r2, r12} can be done:

```
pek = DifferentialSubstitute[pek1[3],
  {r1 == Sqrt[Sum[x[i]^2, {i, 3}]],
   r2 == Sqrt[Sum[y[i]^2, {i, 3}]],
   r12 == Sqrt[Sum[(y[i] - x[i])^2, {i, 3}]]}]
```

$$\mathrm{r1\,r12\,r2}\,D_{r1}^2 + \left(\mathrm{r1}^2\,\mathrm{r2} + \mathrm{r12}^2\,\mathrm{r2} - \mathrm{r2}^3\right) D_{r1}\,D_{r12} + \mathrm{r1\,r12\,r2}\,D_{r2}^2 + \left(-\mathrm{r1}^3 + \mathrm{r1\,r12}^2 + \mathrm{r1\,r2}^2\right) D_{r2}\,D_{r12} + 2\,\mathrm{r1\,r12\,r2}\,D_{r12}^2 + 2\,\mathrm{r12\,r2}\,D_{r1} + 2\,\mathrm{r1\,r12}\,D_{r2} + 4\,\mathrm{r1\,r2}\,D_{r12} + \left(-2\,\mathrm{r1\,r2} + 2\,\mathbb{e}\,\mathrm{r1\,r12\,r2} + 2\,\mathrm{r1\,r12}\,Z + 2\,\mathrm{r12\,r2}\,Z\right)$$

The result agrees with formula (5) in the paper [1].

```
pek - NormalizeCoefficients[pek5]
```

0

Now the energy parameter E is replaced.

```
pek = pek /. E → -e^2;
```

The change of variables to perimetric coordinates can also be obtained by the above procedure DifferentialSubstitute:

```
pek = DifferentialSubstitute[pek,
  {u == e * (r2 + r12 - r1), v == e * (r1 + r12 - r2), w == 2 * e * (r1 + r2 - r12)}]
```

$$\left(-8\,e\,u^2\,v - 8\,e\,u\,v^2 - 8\,e\,u^2\,w - 8\,e\,u\,v\,w - 4\,e\,u\,w^2\right) D_u^2 + \left(16\,e\,u^2\,w + 8\,e\,u\,w^2\right) D_u\,D_w + \left(-8\,e\,u^2\,v - 8\,e\,u\,v^2 - 8\,e\,u\,v\,w - 8\,e\,v^2\,w - 4\,e\,v\,w^2\right) D_v^2 + \left(16\,e\,v^2\,w + 8\,e\,v\,w^2\right) D_v\,D_w + \left(-16\,e\,u^2\,w - 16\,e\,v^2\,w - 8\,e\,u\,w^2 - 8\,e\,v\,w^2\right) D_w^2 + \left(8\,e\,u^2 - 16\,e\,u\,v - 8\,e\,v^2 - 8\,e\,u\,w - 8\,e\,v\,w - 4\,e\,w^2\right) D_u + \left(-8\,e\,u^2 - 16\,e\,u\,v + 8\,e\,v^2 - 8\,e\,u\,w - 8\,e\,v\,w - 4\,e\,w^2\right) D_v + \left(-16\,e\,u^2 - 16\,e\,v^2 + 8\,e\,w^2\right) D_w + \left(8\,u\,v + 4\,e\,u^2\,v + 4\,e\,u\,v^2 + 4\,u\,w + 2\,e\,u^2\,w + 4\,v\,w + 4\,e\,u\,v\,w + 2\,e\,v^2\,w + 2\,w^2 + e\,u\,w^2 + e\,v\,w^2 - 8\,u^2\,Z - 16\,u\,v\,Z - 8\,v^2\,Z - 8\,u\,w\,Z - 8\,v\,w\,Z\right)$$

The package HolonomicFunctions can be used to perform the transformation of the differential equation in order to incorporate the exponential part of the solution. The above differential operator annihilates ψ and using the closure property product, we get an operator for exp((u+v+w)/2)*ψ :

```
pek = First[DFiniteTimes[{pek}, Annihilator[Exp[(u + v + w) / 2], {Der[u], Der[v], Der[w]}]]]
```

$$\left(4 e u^2 v + 4 e u v^2 + 4 e u^2 w + 4 e u v w + 2 e u w^2\right) D_u^2 + \left(-8 e u^2 w - 4 e u w^2\right) D_u D_w + \left(4 e u^2 v + 4 e u v^2 + 4 e u v w + 4 e v^2 w + 2 e v w^2\right) D_v^2 + \left(-8 e v^2 w - 4 e v w^2\right) D_v D_w + \left(8 e u^2 w + 8 e v^2 w + 4 e u w^2 + 4 e v w^2\right) D_w^2 + \left(-4 e u^2 + 8 e u v - 4 e u^2 v + 4 e v^2 - 4 e u v^2 + 4 e u w + 4 e v w - 4 e u v w + 2 e w^2\right) D_u + \left(4 e u^2 + 8 e u v - 4 e u^2 v - 4 e v^2 - 4 e u v^2 + 4 e u w + 4 e v w - 4 e u v w + 2 e w^2\right) D_v + \left(8 e u^2 + 8 e v^2 - 4 e u^2 w - 4 e v^2 w - 4 e w^2 - 2 e u w^2 - 2 e v w^2\right) D_w + \left(-4 e u^2 - 4 u v - 8 e u v - 4 e v^2 - 2 u w - 4 e u w - 2 v w - 4 e v w - w^2 + 4 u^2 Z + 8 u v Z + 4 v^2 Z + 4 u w Z + 4 v w Z\right)$$

Again, our result agrees with formula (14) in [1]:

```
pek - pek14
```

0

Now the differential equation has to be transformed into a recurrence for the coefficients of the power series expansion with respect to Laguerre polynomials. This can be done by reduction with Gröbner bases. For this purpose, first an annihilating ideal for the product of Laguerre polynomials is computed

```
Timing[ann = Annihilator[LaguerreL[l, u] * LaguerreL[m, v] * LaguerreL[n, w],
   {Der[u], Der[v], Der[w], S[l], S[m], S[n]}, MonomialOrder → EliminationOrder[3]]]
```

$$\left\{1.5961, \left\{(2+n) S_n^2 + (-3 - 2 n + w) S_n + (1+n), (2+m) S_m^2 + (-3 - 2 m + v) S_m + (1+m), (2+l) S_l^2 + (-3 - 2 l + u) S_l + (1+l), w D_w + (-1-n) S_n + (1+n-w), v D_v + (-1-m) S_m + (1+m-v), u D_u + (-1-l) S_l + (1+l-u)\right\}\right\}$$

```
(* a faster variant, giving the same result *)
Timing[ann = SortBy[ToOrePolynomial[
    Join @@ (Annihilator[LaguerreL @@ #, {Der[#[[2]]], S[#[[1]]]}] & /@ {{l, u}, {m, v}, {n, w}}),
    OreAlgebra[Der[u], Der[v], Der[w], S[l], S[m], S[n]],
    MonomialOrder → EliminationOrder[3]], LeadingExponent]]
```

$$\left\{0.248016, \left\{(2+n) S_n^2 + (-3 - 2 n + w) S_n + (1+n), (2+m) S_m^2 + (-3 - 2 m + v) S_m + (1+m), (2+l) S_l^2 + (-3 - 2 l + u) S_l + (1+l), w D_w + (-1-n) S_n + (1+n-w), v D_v + (-1-m) S_m + (1+m-v), u D_u + (-1-l) S_l + (1+l-u)\right\}\right\}$$

The following reduction rewrites all D's in terms of shifts:

```
pek = OreReduce[pek, ann]
```

$$(-8 e u - 8 e l u - 8 e n u - 8 e l n u - 4 e w - 4 e l w - 4 e n w - 4 e l n w) S_l S_n + (-8 e v - 8 e m v - 8 e n v - 8 e m n v - 4 e w - 4 e m w - 4 e n w - 4 e m n w) S_m S_n + \left(4 e u + 4 e l u + 8 e n u + 8 e l n u + 4 e v + 4 e l v + 4 e w + 4 e l w + 4 e n w + 4 e l n w - 4 e u w - 4 e l u w - 2 e w^2 - 2 e l w^2\right) S_l + \left(4 e u + 4 e m u + 4 e v + 4 e m v + 8 e n v + 8 e m n v + 4 e w + 4 e m w + 4 e n w + 4 e m n w - 4 e v w - 4 e m v w - 2 e w^2 - 2 e m w^2\right) S_m + \left(4 e u + 8 e l u + 4 e n u + 8 e l n u - 4 e u^2 - 4 e n u^2 + 4 e v + 8 e m v + 4 e n v + 8 e m n v - 4 e v^2 - 4 e n v^2 + 4 e w + 4 e l w + 4 e m w + 4 e n w + 4 e l n w + 4 e m n w - 2 e u w - 2 e n u w - 2 e v w - 2 e n v w\right) S_n + \left(-4 e u - 4 e l u - 4 e m u - 4 e n u - 8 e l n u - 4 e u^2 - 4 e m u^2 - 4 e n u^2 - 4 e v - 4 e l v - 4 e m v - 4 e n v - 8 e m n v - 4 u v - 4 e l u v - 4 e m u v - 4 e v^2 - 4 e l v^2 - 4 e n v^2 - 4 e w - 4 e l w - 4 e m w - 4 e n w - 4 e l n w - 4 e m n w - 2 u w - 2 e u w - 4 e m u w - 2 e n u w - 2 v w - 2 e v w - 4 e l v w - 2 e n v w - w^2 + 4 u^2 Z + 8 u v Z + 4 v^2 Z + 4 u w Z + 4 v w Z\right)$$

Next, the variables u,v,w have to be eliminated. Again, this can be obtained by means of Gröbner bases. An annihilating ideal w.r.t. the shift operators is computed, including negative shifts:

```
ann = Annihilator[LaguerreL[l, u] * LaguerreL[m, v] * LaguerreL[n, w], {S[l], S[m], S[n]}]
```

```
{(2 + n) S_n^2 + (-3 - 2 n + w) S_n + (1 + n),
 (2 + m) S_m^2 + (-3 - 2 m + v) S_m + (1 + m), (2 + l) S_l^2 + (-3 - 2 l + u) S_l + (1 + l)}
```

```
ann = OreGroebnerBasis[Join[ann, {S[l] * S[-1][l] - 1, S[m] * S[-1][m] - 1, S[n] * S[-1][n] - 1}],
  OreAlgebra[u, v, w, S[l], S[m], S[n], S[-1][l], S[-1][m], S[-1][n]],
  MonomialOrder → EliminationOrder[3]]
```

```
{-S_n S[-1][n] + 1, -S_m S[-1][m] + 1, -S_l S[-1][l] + 1, -w + (-1 - n) S_n - n S[-1][n] + (1 + 2 n),
 -v + (-1 - m) S_m - m S[-1][m] + (1 + 2 m), -u + (-1 - l) S_l - l S[-1][l] + (1 + 2 l)}
```

```
op22 = OreReduce[pek, ann];
```

This operator now represents the 33-term recurrence that appears as formula (22) in [1].

```
Length[Support[op22]]
```

```
33
```

Or, in a more human readable format:

```
rel22 = Collect[ApplyOreOperator[op22, A[l, m, n]], A[__], Simplify]
```

```
-4 (-1 + l) l (e (1 + m + n) - Z) A[-2 + l, m, n] - 4 e (-1 + l) l (1 + n) A[-2 + l, m, 1 + n] -
 4 l m (1 + e (l + m) - 2 Z) A[-1 + l, -1 + m, n] - 2 l n (1 + e (1 + 2 m + n) - 2 Z) A[-1 + l, m, -1 + n] +
 2 l (3 + 2 n + e (2 + 6 m + 4 m^2 + 3 n + 4 m n + n^2 + 12 l (1 + m + n)) + m (4 - 8 Z) - 6 Z - 8 l Z - 4 n Z)
   A[-1 + l, m, n] + 2 l (1 + n) (-1 + e (4 l - 2 m + n) + 2 Z) A[-1 + l, m, 1 + n] -
 2 e l (2 + 3 n + n^2) A[-1 + l, m, 2 + n] - 4 l (1 + m) (1 + e (1 + l + m) - 2 Z) A[-1 + l, 1 + m, n] -
 4 (-1 + m) m (e (1 + l + n) - Z) A[l, -2 + m, n] - 4 e (-1 + m) m (1 + n) A[l, -2 + m, 1 + n] -
 2 m n (1 + e (1 + 2 l + n) - 2 Z) A[l, -1 + m, -1 + n] +
 2 m (3 + 2 n + e (4 l^2 + (1 + n) (2 + 12 m + n) + 2 l (3 + 6 m + 2 n)) + l (4 - 8 Z) - 6 Z - 8 m Z - 4 n Z)
   A[l, -1 + m, n] + 2 m (1 + n) (-1 + e (-2 l + 4 m + n) + 2 Z) A[l, -1 + m, 1 + n] -
 2 e m (2 + 3 n + n^2) A[l, -1 + m, 2 + n] - (-1 + n) n A[l, m, -2 + n] +
 4 n (1 - e l^2 - e m^2 + n + 2 e n + m (1 + e + 2 e n - 2 Z) + l (1 + e + 4 e m + 2 e n - 2 Z) - 2 Z) A[l, m, -1 + n] -
 2 (5 + 6 m + 7 n + 4 m n + 3 n^2 + 4 e (5 m^2 (1 + n) + 5 l^2 (1 + m + n) +
        2 (2 + 3 n + n^2) + m (8 + 9 n + 2 n^2) + l (8 + 5 m^2 + 9 n + 2 n^2 + 4 m (3 + n))) +
     l (6 + m (8 - 16 Z) + n (4 - 8 Z) - 24 Z) - 16 Z - 12 l^2 Z - 24 m Z - 12 m^2 Z - 8 n Z - 8 m n Z) A[l, m, n] +
 4 (1 + n) (2 + l + m + n + e (2 - l^2 + 3 m - m^2 + 2 n + 2 m n + l (3 + 4 m + 2 n)) - 2 Z - 2 l Z - 2 m Z) A[l, m, 1 + n] +
 (-2 - 3 n - n^2) A[l, m, 2 + n] - 2 e (1 + m) (-1 + n) n A[l, 1 + m, -2 + n] +
 2 (1 + m) n (-1 + e (3 - 2 l + 4 m + n) + 2 Z) A[l, 1 + m, -1 + n] +
 2 (1 + m) (3 + 2 n + e (4 l^2 + (1 + n) (14 + 12 m + n) + 2 l (9 + 6 m + 2 n)) + l (4 - 8 Z) - 14 Z - 8 m Z - 4 n Z)
   A[l, 1 + m, n] - 2 (1 + m) (1 + n) (1 + e (2 + 2 l + n) - 2 Z) A[l, 1 + m, 1 + n] -
 4 e (2 + 3 m + m^2) n A[l, 2 + m, -1 + n] - 4 (2 + 3 m + m^2) (e (1 + l + n) - Z) A[l, 2 + m, n] -
 4 (1 + l) m (1 + e (1 + l + m) - 2 Z) A[1 + l, -1 + m, n] - 2 e (1 + l) (-1 + n) n A[1 + l, m, -2 + n] +
 2 (1 + l) n (-1 + e (3 + 4 l - 2 m + n) + 2 Z) A[1 + l, m, -1 + n] +
 2 (1 + l) (3 + 2 n + e (14 + 18 m + 4 m^2 + 15 n + 4 m n + n^2 + 12 l (1 + m + n)) + m (4 - 8 Z) - 14 Z - 8 l Z - 4 n Z)
   A[1 + l, m, n] - 2 (1 + l) (1 + n) (1 + e (2 + 2 m + n) - 2 Z) A[1 + l, m, 1 + n] -
 4 (1 + l) (1 + m) (1 + e (2 + l + m) - 2 Z) A[1 + l, 1 + m, n] - 4 e (2 + 3 l + l^2) n A[2 + l, m, -1 + n] -
 4 (2 + 3 l + l^2) (e (1 + m + n) - Z) A[2 + l, m, n]
```

The recurrence is self-adjoint and therefore we are done. This is checked by verifying Eq. (24) in [1]:

```
Union[Flatten[Expand[Table[Coefficient[op22, S[-1][l]^a * S[-1][m]^b * S[-1][n]^c] -
    (Coefficient[op22, S[l]^a * S[m]^b * S[n]^c] /. l → l - a /. m → m - b /. n → n - c),
   {a, 0, 2}, {b, 0, 2}, {c, 0, 2}]]]]
```

```
{0}
```

Matrix in the symmetric case (para)

We define the scheme that is used for ordering the indices l,m,n (formula (28), corrected).

```
SchemePara[l_Integer /; l ≥ 0, m_Integer /; m ≥ 0, n_Integer /; n ≥ 0] :=
  With[{w = l + m + n}, (1 / 24) * w * (w + 2) * (2 w + 5) + (1 / 16) * (1 - (-1) ^w) +
    (1 / 4) * (l + m) ^2 + (1 / 8) * (1 - (-1) ^ (l + m)) + (l + m) / 2 + l + 1];
```

The following procedure generates the matrix for the symmetric ("para") case: A(l,m,n) = A(m,l,n).

```
MatrixPara[rel_, A_[l_, m_, n_], omega_Integer] :=
  Module[{a1, a2, a3, a, c, t, l1, lm, w, dim, rel2, row, matrix = {}},
   dim = Ceiling[(omega + 1) * (omega + 3) * (2 * omega + 7) / 24];
   Do[
    rel2 = rel /. {l → l1, m → lm - l1, n → w - lm}
       /. A[a1_, a2_, a3_] /; a1 > a2 → A[a2, a1, a3]
       /. A[a1_, a2_, a3_] /; a1 + a2 + a3 > omega → 0;
    rel2 = If[Head[rel2] === Plus, List @@ rel2, {rel2}];
    row = Table[0, {dim}];
    Function[t,
      a = First[Cases[t, A[__], Infinity]];
      c = t /. a → 1;
      row[[SchemePara @@ a]] += c;] /@ rel2;
    AppendTo[matrix, row];
    , {w, 0, omega}, {lm, 0, w}, {l1, 0, lm / 2}];
   Return[matrix];
  ];
```

```
TableForm[MatrixPara[rel22, A[l, m, n], 1]]
```

-2 (5 + 16 e - 16 Z)	4 (2 + 2 e - 2 Z)	4 (3 + 14 e - 14 Z)
4 (2 + 2 e - 2 Z)	-2 (15 + 48 e - 24 Z)	4 (-1 + 4 e + 2 Z)
2 (3 + 14 e - 14 Z)	2 (-1 + 4 e + 2 Z)	-2 (11 + 68 e - 52 Z) - 4 (1 + 2 e - 2 Z)

We now can reproduce Table II in [1] (p. 1651). The rows are normalized as to make the matrix symmetric.

```
TableForm[Expand[MatrixPara[-rel22, A[l, m, n], 2] / {2, 2, 1, 2, 1, 1, 2}]]
```

5 + 16 e - 16 Z	-4 - 4 e + 4 Z	-6 - 28 e + 28 Z	1	2 + 4 e - 4 Z	8 e - 8 Z	2 +
-4 - 4 e + 4 Z	15 + 48 e - 24 Z	2 - 8 e - 4 Z	-12 - 16 e + 8 Z	-10 - 60 e + 36 Z	8 e	0
-6 - 28 e + 28 Z	2 - 8 e - 4 Z	26 + 144 e - 112 Z	4 e	-12 - 16 e + 16 Z	-12 - 104 e + 88 Z	-14
1	-12 - 16 e + 8 Z	4 e	31 + 96 e - 32 Z	4 - 20 e - 8 Z	0	0
2 + 4 e - 4 Z	-10 - 60 e + 36 Z	-12 - 16 e + 16 Z	4 - 20 e - 8 Z	54 + 336 e - 144 Z	4 - 32 e - 8 Z	2 -
8 e - 8 Z	8 e	-12 - 104 e + 88 Z	0	4 - 32 e - 8 Z	34 + 320 e - 224 Z	8 +
2 + 4 e - 4 Z	0	-14 - 72 e + 44 Z	0	2 - 4 e - 4 Z	8 + 24 e - 16 Z	25

Matrix in the antisymmetric case (ortho)

We define the scheme that is used for ordering the indices l,m,n (formula (28), corrected).

```
SchemeOrtho[l_Integer /; l ≥ 0, m_Integer /; m ≥ 0, n_Integer /; n ≥ 0] :=
  With[{w = l + m + n}, (1 / 24) * w * (w + 2) * (2 w - 1) - (1 / 16) * (1 - (-1) ^w) + l * (m + n) + m];
```

The following procedure generates the matrix for the antisymmetric ("ortho") case: A(l,m,n) = -A(m,l,n).

```
MatrixOrtho[rel_, A[l_, m_, n_], omega_Integer] :=
  Module[{a1, a2, a3, a, c, t, l1, m1, w, dim, rel2, row, matrix = {}},
   dim = Floor[(omega + 1) * (omega + 3) * (2 * omega + 1) / 24];
   Do[
    rel2 = rel /. {l → l1, m → m1, n → w - l1 - m1}
        /. A[a1_, a2_, a3_] /; a1 > a2 → -A[a2, a1, a3]
        /. A[a1_, a1_, a2_] → 0
        /. A[a1_, a2_, a3_] /; a1 + a2 + a3 > omega → 0;
    rel2 = If[Head[rel2] === Plus, List @@ rel2, {rel2}];
    row = Table[0, {dim}];
    Function[t,
      a = First[Cases[t, A[__], Infinity]];
      c = t /. a → 1;
      row[[SchemeOrtho @@ a]] += c;] /@ rel2;
    AppendTo[matrix, row];
    , {w, 0, omega}, {l1, 0, (w - 1) / 2}, {m1, l1 + 1, w - l1}];
   Return[matrix];
  ];
```

We now can reproduce Table IV in [2] (p. 1219):

```
TableForm[Expand[MatrixOrtho[-rel22, A[l, m, n], 3] / 2]]
```

9 + 64 e - 48 Z	-6 - 8 e + 8 Z	-6 - 52 e + 44 Z	1	2 + 4 e - 4 Z	12 e - 12 Z	4
-6 - 8 e + 8 Z	23 + 160 e - 64 Z	2 - 16 e - 4 Z	-16 - 32 e + 16 Z	-10 - 108 e + 52 Z	12 e	-
-6 - 52 e + 44 Z	2 - 16 e - 4 Z	17 + 160 e - 112 Z	4 e	-8 - 8 e + 12 Z	-9 - 114 e + 90 Z	-
1	-16 - 32 e + 16 Z	4 e	43 + 288 e - 80 Z	4 - 36 e - 8 Z	0	0
2 + 4 e - 4 Z	-10 - 108 e + 52 Z	-8 - 8 e + 12 Z	4 - 36 e - 8 Z	35 + 360 e - 136 Z	3 - 36 e - 6 Z	1
12 e - 12 Z	12 e	-9 - 114 e + 90 Z	0	3 - 36 e - 6 Z	23 + 292 e - 196 Z	6
4 + 4 e - 4 Z	-4 e	-11 - 66 e + 30 Z	0	1 - 2 Z	6 + 24 e - 12 Z	3

For relatively small dimensions, the determinant can be evaluated symbolically:

```
Det[MatrixOrtho[rel22 /. Z → 2, A[l, m, n], 4]]
```

$$-47\,775\,744\,\big(-21\,524\,781\,278\,029\,797\,978\,561\,675 + 512\,411\,332\,823\,022\,623\,074\,122\,240\,e - 5\,466\,482\,266\,891\,795\,745\,966\,559\,360\,e^2 + 34\,677\,367\,119\,795\,181\,663\,130\,869\,248\,e^3 - 146\,180\,473\,643\,682\,656\,469\,493\,203\,328\,e^4 + 432\,948\,975\,539\,938\,537\,427\,450\,884\,096\,e^5 - 927\,726\,977\,944\,633\,329\,428\,839\,233\,536\,e^6 + 1\,457\,213\,342\,415\,069\,436\,091\,637\,465\,088\,e^7 - 1\,678\,589\,828\,014\,516\,496\,523\,878\,694\,912\,e^8 + 1\,400\,928\,650\,920\,504\,827\,263\,329\,501\,184\,e^9 - 823\,517\,769\,545\,976\,023\,048\,860\,467\,200\,e^{10} + 322\,908\,295\,237\,814\,393\,725\,438\,132\,224\,e^{11} - 75\,698\,939\,730\,253\,386\,299\,253\,719\,040\,e^{12} + 8\,012\,928\,704\,859\,449\,627\,089\,305\,600\,e^{13}\big)$$

```
NSolve[% == 0, e]
```

```
{{e → 0.210548}, {e → 0.310105}, {e → 0.446592}, {e → 0.459615},
 {e → 0.485099}, {e → 0.6328}, {e → 0.641195}, {e → 0.73747},
 {e → 0.763258}, {e → 0.800272}, {e → 1.13401}, {e → 1.3563}, {e → 1.46983}}
```

A simple procedure using chinese remaindering and polynomial interpolation can symbolically evaluate such determinants much faster:

```
(* it is assumed that the coefficients are integers. *)
DetUniv[mat_] :=
  Module[{deg, var, det, det1, i, p, pp, vals, len, mats, xs},
   var = Variables[mat];
   If[Length[var] === 1, var = First[var],
    Throw["matrix entries must be univariate polynomials."]];
   deg = Total[Function[row, Max[Exponent[#, var] & /@ row]] /@ mat];
   mats = Table[mat /. var → i, {i, Ceiling[-deg / 2], Ceiling[deg / 2]}];
   xs = Range[Ceiling[-deg / 2], Ceiling[deg / 2]];
   det = {}; det1 = Null;
   p = Developer`$MaxMachineInteger; pp = 1;
   While[det =!= det1,
    p = NextPrime[p, -1];
    vals = {};
    det1 = det;
    vals = Transpose[{xs, Det[#, Modulus → p] & /@ mats}];
    det = Mod[CoefficientList[InterpolatingPolynomial[vals, var, Modulus → p], var], p];
    If[det1 =!= {},
     len = Max[Length /@ {det, det1}];
     {det, det1} = PadRight[#, len] & /@ {det, det1};
     det = ChineseRemainder[#, {pp, p}] & /@ Transpose[{det1, det}];
    ];
    pp *= p;
    det = det /. a_Integer /; a > pp / 2 :→ a - pp;
   ];
   Return[det.Table[var^i, {i, 0, Length[det] - 1}]];
  ];
```

```
Timing[
 Last[NSolve[Det[MatrixOrtho[rel22 /. Z → 2, A[l, m, n], 10]] == 0, e, WorkingPrecision → 20]]]
```

```
{1397.54, {e → 1.4748630375781090750}}
```

```
Timing[Last[
  NSolve[DetUniv[MatrixOrtho[rel22 /. Z → 2, A[l, m, n], 10]] == 0, e, WorkingPrecision → 20]]]
```

```
{126.856, {e → 1.4748630375781090750}}
```

Generate Tables

We can generate (with symbolic determinants!) the first two columns of Table III on page 1652 in [1].

```
Timing[tab = Table[
    Last[NSolve[DetUniv[MatrixPara[rel22, A[l, m, n], om]] == 0, e, WorkingPrecision → 12]][[
     1, 2]], {Z, 10}, {om, 8, 9}];]
```

```
{1594.94, Null}
```

```
TableForm[tab]
```

0.726464050394	0.726464459043
1.70403151059	1.70403165423
2.69813125517	2.69813136130
3.69534369533	3.69534377798
4.69371604698	4.69371611435
5.69264835790	5.69264841466
6.69189389162	6.69189394062
7.69133238251	7.69133242560
8.69089817280	8.69089821125
9.69055236783	9.69055240253

For generating the values in the tables, the symbolic evaluation of the determinants is too slow. Therefore we use numeric methods which can be applied when the problem is reformulated as to find the generalized eigenvalues of the following two matrices:

```
mat = MatrixPara[rel22 /. Z → 2, A[l, m, n], 8];
mA = mat /. e → 0;
mB = -(mat - mA) /. e → 1;
```

```
Eigenvalues[N[{mA, mB}], 1]
```

```
{1.70403}
```

Unfortunately, Mathematica does not allow for more digits:

```
Eigenvalues[N[{mA, mB}, 20], 1];
```

```
Eigenvalues::gfargs :
 Generalized Eigenvalues arguments accept only matrices with machine
     real and complex elements. >>
```

Therefore, we will perform these numeric computations in Matlab, too (with higher precision and much faster than below).

```
PekerisTable[rel_, A_[l_, m_, n_], case_String, oms_List, zs_List] :=
  Module[{mat, mat1, matf, omi, om, zi, matA, matB, ev, col, table},
   matf = Switch[case, "para", MatrixPara, "ortho", MatrixOrtho, _,
     Throw["Unknown case. Give either \"para\" or \"ortho\"."]];
   table = Join[{{"\\ ω"}, {"Z\\n"}, {"---"}}, Transpose[{zs}]];
   Do[
    om = oms[[omi]];
    mat = If[Length[zs] === 1,
      matf[rel /. Z → First[zs], A[l, m, n], om], matf[rel, A[l, m, n], om]];
    col = {om, First[Dimensions[mat]], "--------"};
    Do[
     mat1 = If[Length[zs] > 1, mat /. Z → zs[[zi]], mat];
     matA = mat1 /. e → 0;
     matB = -(mat1 - matA) /. e → 1;
     ev = Eigenvalues[N[{matA, matB}], 1];
     AppendTo[col, ev];
     , {zi, Length[zs]}];
    table = MapThread[Append, {table, col}];
    , {omi, Length[oms]}];
   Return[table];
  ];
```

Table III on page 1652 in [1].

```
Timing[tab1 = PekerisTable[rel22, A[l, m, n], "para", Range[8, 11], Range[10]];]
```

```
{10.9967, Null}
```

```
TableForm[tab1]
```

\ ω	8	9	10	11
Z\n	95	125	161	203
1	0.726464	0.726464	0.726465	0.726465
2	1.70403	1.70403	1.70403	1.70403
3	2.69813	2.69813	2.69813	2.69813
4	3.69534	3.69534	3.69534	3.69534
5	4.69372	4.69372	4.69372	4.69372
6	5.69265	5.69265	5.69265	5.69265
7	6.69189	6.69189	6.69189	6.69189
8	7.69133	7.69133	7.69133	7.69133
9	8.6909	8.6909	8.6909	8.6909
10	9.69055	9.69055	9.69055	9.69055

Table I on page 1217 in [2].

```
Timing[tab2 = PekerisTable[rel22, A[l, m, n], "para", Range[12, 21, 3], {2}];]
```

```
{74.8447, Null}
```

```
TableForm[tab2]
```

\ ω	12	15	18	21
Z\n	252	444	715	1078
2	1.70403	1.70403	1.70403	1.70403

Table V on page 1220 in [2].

```
Timing[tab3 = PekerisTable[rel22, A[l, m, n], "ortho", Range[10, 19, 3], {2}];]
```

```
{15.481, Null}
```

```
TableForm[tab3]
```

\ ω	10	13	16	19
Z\n	125	252	444	715
2	1.47486	1.47487	1.47487	1.47487